// File: contracts/GalverseKumade.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity 0.8.17;

import "./Kumade.sol";

///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//                                                                                                                   //
//                                                                                                                   //
//                                    _8888_                      ▄▀▀▀▀▀▌                                            //
//                            ,▀▀▀▀▀▀▀      ▀▀▀▀▀▀▀▀▌ █▀▀▀▀▀▀▀▀▀▀▀     █░                                            //
//                           █▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄'░█               ▄█░                                             //
//                             ░░░░░░░░░░░░░░░░░░░░░█   ,████████████▀░                                              //
//                         █▀▀▀▀▀▄       ▄▀▀▀▀▀█   █   ,█░░░░░░░░░░░░░                                               //
//                          █     ▀▄▄▄▄▄▀    ▄█   █    ▀▄▄▄▄▄▄▄▄▄▄▄▄▄▄_                                              //
//                       ,▄▄▄█              █▄▄▄ █                  █░                                               //
//                      █                     █░█     █████     ,███░                                                //
//                     ²████████▀    ╓████████░█     █▀░░█     █▀▀░                                                  //
//                       ░░░░░░█    ,█░░░░░░░░█     █▀░ ▐     ██░                                                    //
//                     ,▄▄▄▄▄▄█     █▄▄▄▄▄▄▄▄▄_    █▀░ ▐     ██░   █▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀╗           //
//                    ▐                     █▀    █▀░ ▐     ██░   █                                    ██░           //
//                   ²███████▀    ╓█████████░    █▀░ ▐     ██░   █     ,████████████████████████Γ     ██░            //
//                     ░░░░░▀    █░░░░░░░░░█    █▀░ ▐     ██░   █     Æ░░░░░░░░░░░░░░░░░░░░░░░░'     ██░             //
//                  ,▄▄▄▄▄▄█    █▄▄▄▄▄▄   █    █▀░ ▐     ██░   █                                    ██░              //
//                 Æ     ▄▄    ▄▄    █░  █    █▀░ ▐     ██░   █     ,████████████████████████Γ     ██░               //
//               Æ'    ,██¬    ▌░    █░ █    █▀░ ▐     ██░   █      Æ░░░░░░░░░░░░░░░░░░░░░░░'     ██░                //
//             ▄█     ▄██¬    ▌░░▌   └█     █▀░ ▐     ██░   █                                   ▄██░                 //
//            ▌     ▄█░█¬    █░░ ▌        ▄█▀░ ▐     ██░   █████████████████     ╓███████████████▀░                  //
//            ▌▄▄▄▄█░░███████░░   ████████▀░░ ▌███████░     ░░░░░░░░░░░░░░█     █░░░░░░░░░░░░░░░░░                   //
//              ░░░░░   ░░░░░░     ░░░░░░░░░     ░░░░░ ___,▀    j▀▀▀▀▀▀▀▀▀     ▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀█                   //
//                                                    █                                         █░                   //
//                                                   █_      ,▄██████████▀    ╓████████████████▀░                    //
//                                                     █    '░░░░░░░░░░░▄    ,█░░░░░░░░░░░░░░░░░                     //
//                                                    █     ▄▄▄▄▄▄▄▄▄▄▄█    '█▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄_                     //
//                                                   █                                       █░                      //
//                                                   █████████████████▀     ╓███████████████▀░                       //
//                                                      ░░░░░░░░░░░░░▀     ,█▌░░░░░░░░░░░░░░░                        //
//                                                █▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀▀█     '█▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄_                        //
//                                               █                                        █░                         //
//                                              ²████████████████████████████████████████▀░                          //
//                                                ░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░                           //
//       — Emi, Ayaka, Devin & Jack                                                                                  //
//                                                                                                                   //
///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

contract GalverseKumade is Kumade {
    string _name = "Shinsei Galverse";
    string _symbol = "Shinsei Galverse";

    constructor() Kumade(_name, _symbol) {}
}

// File: contracts/Kumade.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity >=0.8.13;

import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
import "@openzeppelin/contracts/token/ERC721/extensions/ERC721Burnable.sol";
import "@openzeppelin/contracts/token/ERC721/extensions/ERC721URIStorage.sol";
import "@openzeppelin/contracts/access/AccessControl.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

import "@openzeppelin/contracts/utils/Strings.sol";
import "./CollectionManager.sol";
import "./interfaces/IERC2981.sol";
import "./interfaces/IOperatorFilterRegistry.sol";

/// @title Kumade Generic Multi-Collection ERC721
contract Kumade is
    ERC721,
    ERC721Burnable,
    ERC721URIStorage,
    AccessControl,
    CollectionManager,
    IERC2981,
    Ownable
{
    using Strings for uint256;
    uint256 public totalSupply;
    mapping(uint256 => address) public originalOwner;

    constructor(
        string memory _name,
        string memory _symbol
    ) ERC721(_name, _symbol) {
        _setupRole(DEFAULT_ADMIN_ROLE, msg.sender);
    }

    /// @dev Precondition that the caller is an admin or the artist
    modifier onlyCollectionAdmin(uint256 collectionId) override {
        require(
            hasRole(DEFAULT_ADMIN_ROLE, msg.sender) ||
                (collectionId < collections.length &&
                    msg.sender == collections[collectionId].artist),
            "not authorized"
        );
        _;
    }

    /// @dev Precondition that a quantity is not zero
    modifier quantityNonZero(uint256 quantity) {
        require(quantity > 0, "quantity is zero");
        _;
    }

    /// @dev Precondition that the collection is active
    modifier collectionActive(uint256 collectionId) {
        require(collections[collectionId].active, "collection not active");
        _;
    }

    /// @dev Precondition that the collection has started
    modifier collectionStarted(uint256 collectionId) {
        require(
            collections[collectionId].startTime <= block.timestamp,
            "collection not started"
        );
        _;
    }

    /// @dev Precondition that the correct payment is attached
    modifier collectionPaymentAttached(uint256 collectionId) {
        require(
            collections[collectionId].price == msg.value,
            "incorrect payment"
        );
        _;
    }

    /// @dev Precondition that the correct payment is attached
    modifier collectionFriendPaymentAttached(uint256 collectionId) {
        require(
            collections[collectionId].friendPrice == msg.value,
            "incorrect payment"
        );
        _;
    }

    /// @dev Precondition that the payment receiver exists
    modifier collectionPaymentReceiverExists(uint256 collectionId) {
        require(
            collections[collectionId].paymentReceiver != address(0),
            "no payment receiver"
        );
        _;
    }

    /// @notice Mint tokens from a collection to the caller
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function safeMint(
        uint256 collectionId,
        string calldata uri
    ) public payable {
        safeMintTo(msg.sender, collectionId, uri);
    }

    /// @notice Mint tokens from a collection to an account
    /// @param to The account to mint to
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function safeMintTo(
        address to,
        uint256 collectionId,
        string calldata uri
    )
        public
        payable
        collectionExists(collectionId)
        collectionActive(collectionId)
        collectionStarted(collectionId)
        collectionPaymentAttached(collectionId)
        collectionPaymentReceiverExists(collectionId)
    {
        _safeMintTo(to, collectionId, uri);
        (bool ok, ) = payable(collections[collectionId].paymentReceiver).call{
            value: msg.value
        }("");
        require(ok, "payment failed");
    }

    /// @notice Mint tokens from a collection to the caller, who must be the
    /// collection artist
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function artistMint(uint256 collectionId, string calldata uri) public {
        artistMintTo(msg.sender, collectionId, uri);
    }

    /// @notice Mint tokens from a collection to an account, must be called by
    /// the collection artist
    /// @param to The account to mint to
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function artistMintTo(
        address to,
        uint256 collectionId,
        string calldata uri
    ) public onlyCollectionAdmin(collectionId) collectionExists(collectionId) {
        _safeMintTo(to, collectionId, uri);
    }

    /// @notice Mint tokens from a collection to an account, must be called by
    /// the collection artist
    /// @param to The account to mint to
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function artistMultiMintTo(
        address[] calldata to,
        uint256 collectionId,
        string[] calldata uri
    )
        public
        quantityNonZero(to.length)
        onlyCollectionAdmin(collectionId)
        collectionExists(collectionId)
    {
        _safeMintTos(to, collectionId, uri);
    }

    /// @notice Mint friend tokens from a collection to an account
    /// @param to The account to mint to
    /// @param collectionId The id of the collection
    /// @param originalTokenId The original token id to base the friend mint on
    /// @param uri The Token URI
    function safeFriendMintTo(
        address to,
        uint256 collectionId,
        uint256 originalTokenId,
        string calldata uri
    )
        public
        payable
        collectionExists(collectionId)
        collectionActive(collectionId)
        collectionStarted(collectionId)
        collectionFriendPaymentAttached(collectionId)
        collectionPaymentReceiverExists(collectionId)
    {
        require(_exists(originalTokenId), "nonexistent token");
        require(
            collectionId == _getCollectionId(originalTokenId),
            "token and collection id mismatch"
        );
        require(
            msg.sender == originalOwner[originalTokenId],
            "msg.sender not the original token owner"
        );

        uint256 tokenId = _consumeTokenIds(collectionId, 1);
        _safeMint(to, tokenId);
        _setTokenURI(tokenId, uri);

        (bool ok, ) = payable(collections[collectionId].paymentReceiver).call{
            value: msg.value
        }("");
        require(ok, "payment failed");
    }

    /// @notice Mint tokens from a collection to an address
    /// @param to The address to mint to
    /// @param collectionId The id of the collection
    /// @param uri The Token URI
    function _safeMintTo(
        address to,
        uint256 collectionId,
        string calldata uri
    ) internal {
        uint256 tokenId = _consumeTokenIds(collectionId, 1);
        address artist = collections[collectionId].artist;

        _safeMintInner(to, collectionId, tokenId, artist, uri);
    }

    /// @notice Mint tokens from a collection to an address
    /// @param tos The addresses to mint to
    /// @param collectionId The id of the collection
    /// @param uris The Token URIs
    function _safeMintTos(
        address[] calldata tos,
        uint256 collectionId,
        string[] calldata uris
    ) internal {
        require(tos.length == uris.length, "length mismatch");

        uint256 firstTokenId = _consumeTokenIds(collectionId, tos.length);
        address artist = collections[collectionId].artist;

        for (uint256 i = 0; i < tos.length; i++) {
            _safeMintInner(
                tos[i],
                collectionId,
                firstTokenId + i,
                artist,
                uris[i]
            );
        }
    }

    /// @notice Set the original owner of a token
    /// @param tokenId The token id
    /// @param to Address to set the original owner to
    function setOriginalOwner(
        uint256 tokenId,
        address to
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        require(
            _exists(tokenId),
            "ERC721URIStorage: URI set of nonexistent token"
        );
        originalOwner[tokenId] = to;
    }

    function _setOriginalOwner(uint256 tokenId, address to) internal {
        originalOwner[tokenId] = to;
    }

    function _safeMintInner(
        address to,
        uint256 collectionId,
        uint256 tokenId,
        address artist,
        string calldata uri
    ) internal {
        bool directMint = collections[collectionId].directMint || to == artist;
        if (directMint) {
            _safeMint(to, tokenId);
        } else {
            _safeMint(artist, tokenId);
            _safeTransfer(artist, to, tokenId, "");
        }
        _setTokenURI(tokenId, uri);
        _setOriginalOwner(tokenId, to);
    }

    function updateTokenURI(
        uint256 tokenId,
        string calldata uri
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        _setTokenURI(tokenId, uri);
    }

    /// @notice Transfer multiple tokens to multiple accounts
    /// @param to The accounts to transfer to
    /// @param tokenIds The respective ids of the tokens to transfer
    function safeBatchTransfer(
        address[] calldata to,
        uint256[] calldata tokenIds
    ) public {
        require(to.length == tokenIds.length, "array lengths must match");
        for (uint256 i = 0; i < to.length; i++) {
            safeTransferFrom(msg.sender, to[i], tokenIds[i]);
        }
    }

    /// @notice Get the id of a collection for a given token
    /// @param tokenId The id of the token
    /// @return The id of the collection
    function getCollectionIdFromTokenId(
        uint256 tokenId
    ) public view returns (uint256) {
        require(_exists(tokenId), "token does not exist");
        return _getCollectionId(tokenId);
    }

    /// @notice See {IERC165-supportsInterface}.
    function supportsInterface(
        bytes4 interfaceId
    ) public view override(ERC721, AccessControl, IERC165) returns (bool) {
        return
            super.supportsInterface(interfaceId) ||
            interfaceId == type(IERC2981).interfaceId;
    }

    /// @notice See {IERC2981-royaltyInfo}
    function royaltyInfo(
        uint256 _tokenId,
        uint256 _salePrice
    ) external view returns (address receiver, uint256 royaltyAmount) {
        require(_exists(_tokenId), "token does not exist");
        uint256 collectionId = _getCollectionId(_tokenId);
        return (
            collections[collectionId].paymentReceiver,
            (collections[collectionId].royaltyPercentage * _salePrice) /
                10 ** 18
        );
    }

    /// @notice Transfers ownership of the contract to a new account (`newOwner`).
    /// @param newOwner New owner
    function transferOwnership(
        address newOwner
    ) public virtual override onlyRole(DEFAULT_ADMIN_ROLE) {
        require(newOwner != address(0), "new owner is the zero address");
        _transferOwnership(newOwner);
    }

    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal override {
        super._beforeTokenTransfer(from, to, tokenId);
        if (from == address(0)) {
            totalSupply++;
        }
        if (to == address(0)) {
            totalSupply--;
        }
    }

    // See: https://github.com/ProjectOpenSea/operator-filter-registry

    IOperatorFilterRegistry public operatorFilterRegistry;

    error OperatorNotAllowed(address operator);

    function subscribeToFilterRegistry(
        address subscriptionOrRegistrantToCopy,
        bool subscribe
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        // If an inheriting token contract is deployed to a network without the registry deployed, the modifier
        // will not revert, but the contract will need to be registered with the registry once it is deployed in
        // order for the modifier to filter addresses.
        if (address(operatorFilterRegistry).code.length > 0) {
            if (subscribe) {
                operatorFilterRegistry.registerAndSubscribe(
                    address(this),
                    subscriptionOrRegistrantToCopy
                );
            } else {
                if (subscriptionOrRegistrantToCopy != address(0)) {
                    operatorFilterRegistry.registerAndCopyEntries(
                        address(this),
                        subscriptionOrRegistrantToCopy
                    );
                } else {
                    operatorFilterRegistry.register(address(this));
                }
            }
        }
    }

    event OperatorFilterRegistrySet(address indexed newOperatorFilterRegistry);

    function setOperatorFilterRegistry(
        address _operatorFilterRegistry
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        operatorFilterRegistry = IOperatorFilterRegistry(
            _operatorFilterRegistry
        );
        emit OperatorFilterRegistrySet(_operatorFilterRegistry);
    }

    modifier onlyAllowedOperator(address from) {
        // Check registry code length to facilitate testing in environments without a deployed registry.
        if (address(operatorFilterRegistry).code.length > 0) {
            // Allow spending tokens from addresses with balance
            // Note that this still allows listings and marketplaces with escrow to transfer tokens if transferred
            // from an EOA.
            if (from == msg.sender) {
                _;
                return;
            }
            if (
                !(operatorFilterRegistry.isOperatorAllowed(
                    address(this),
                    msg.sender
                ) &&
                    operatorFilterRegistry.isOperatorAllowed(
                        address(this),
                        from
                    ))
            ) {
                revert OperatorNotAllowed(msg.sender);
            }
        }
        _;
    }

    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public override onlyAllowedOperator(from) {
        super.transferFrom(from, to, tokenId);
    }

    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public override onlyAllowedOperator(from) {
        super.safeTransferFrom(from, to, tokenId);
    }

    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public override onlyAllowedOperator(from) {
        super.safeTransferFrom(from, to, tokenId, data);
    }

    function _burn(
        uint256 _tokenId
    ) internal override(ERC721URIStorage, ERC721) {
        super._burn(_tokenId);
    }

    function tokenURI(
        uint256 tokenId
    ) public view override(ERC721, ERC721URIStorage) returns (string memory) {
        return super.tokenURI(tokenId);
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/ERC721Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "../../../utils/Context.sol";

/**
 * @title ERC721 Burnable Token
 * @dev ERC721 Token that can be irreversibly burned (destroyed).
 */
abstract contract ERC721Burnable is Context, ERC721 {
    /**
     * @dev Burns `tokenId`. See {ERC721-_burn}.
     *
     * Requirements:
     *
     * - The caller must own `tokenId` or be an approved operator.
     */
    function burn(uint256 tokenId) public virtual {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721Burnable: caller is not owner nor approved");
        _burn(tokenId);
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721URIStorage.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/ERC721URIStorage.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";

/**
 * @dev ERC721 token with storage based token URI management.
 */
abstract contract ERC721URIStorage is ERC721 {
    using Strings for uint256;

    // Optional mapping for token URIs
    mapping(uint256 => string) private _tokenURIs;

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721URIStorage: URI query for nonexistent token");

        string memory _tokenURI = _tokenURIs[tokenId];
        string memory base = _baseURI();

        // If there is no base URI, return the token URI.
        if (bytes(base).length == 0) {
            return _tokenURI;
        }
        // If both are set, concatenate the baseURI and tokenURI (via abi.encodePacked).
        if (bytes(_tokenURI).length > 0) {
            return string(abi.encodePacked(base, _tokenURI));
        }

        return super.tokenURI(tokenId);
    }

    /**
     * @dev Sets `_tokenURI` as the tokenURI of `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _setTokenURI(uint256 tokenId, string memory _tokenURI) internal virtual {
        require(_exists(tokenId), "ERC721URIStorage: URI set of nonexistent token");
        _tokenURIs[tokenId] = _tokenURI;
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual override {
        super._burn(tokenId);

        if (bytes(_tokenURIs[tokenId]).length != 0) {
            delete _tokenURIs[tokenId];
        }
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role, _msgSender());
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: contracts/CollectionManager.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity >=0.8.13;

abstract contract CollectionManager {
    // TYPES -------------------------------------------------------------------

    /// @notice Represents a collection of pieces
    struct Collection {
        /// @notice Artist address
        address artist;
        /// @notice Payment receiver address
        address paymentReceiver;
        /// @notice Price in ETH (wei)
        uint256 price;
        /// @notice Friend price in ETH (wei)
        uint256 friendPrice;
        /// @notice Supply (0 for unlimited, >= 1 for fixed)
        uint128 supply;
        /// @notice Time after which the collection is open for public sale
        uint256 startTime;
        /// @notice Max quantity purchasable per transaction (0 for unlimited)
        uint256 perMintQuantity;
        /// @notice Royalty percentage (0 to 1), 18 decimals
        uint256 royaltyPercentage;
        /// @notice Inactive collections cannot be minted/bought by anyone
        bool active;
        /// @notice If false, then tokens are first minted to the artist's
        /// wallet then transferred to the recipient
        bool directMint;
    }

    /// @notice Contains internal state regarding a collection
    struct CollectionState {
        /// @notice The local ID of the next token in the collection.
        uint256 nextId;
    }

    // STATE -------------------------------------------------------------------

    /// @notice All collections associated with this contract
    Collection[] public collections;

    /// @notice CollectionState for each Collection in collections
    CollectionState[] public collectionStates;

    // MODIFIERS ---------------------------------------------------------------

    /// @dev precondition that the collection must exist
    modifier collectionExists(uint256 collectionId) {
        require(collectionId < collections.length, "invalid collection");
        _;
    }

    /// @dev note: collectionId may be invalid in the case of a new collection
    modifier onlyCollectionAdmin(uint256 collectionId) virtual {
        // Note: These lines do not need to be covered by tests - solidity
        // coverage just fails to compile when a virtual modifier does not have
        // a body.
        require(false, "this is a virtual modifier");
        _;
    }

    // VIEWS -------------------------------------------------------------------

    /// @notice Get the global token id for a given collection and local id
    /// @param collectionId The id of the collection
    /// @param localId The local token id in the collection
    /// @return The global token id for the token in the collection
    function _getGlobalTokenId(
        uint256 collectionId,
        uint256 localId
    ) private pure returns (uint256) {
        return (collectionId << 128) + localId;
    }

    /// @notice Get the id of a collection for a given token id
    /// @param tokenId The id of the token
    /// @return The id of the collection
    function _getCollectionId(uint256 tokenId) internal pure returns (uint256) {
        return tokenId >> 128;
    }

    /// @notice Get the number of collections
    /// @return The number of collections
    function getCollectionCount() external view returns (uint256) {
        return collections.length;
    }

    /// @notice Get the min, next and max token ids for a given collection
    /// @param collectionId The id of the collection
    /// @return min The first token id for the collection
    /// @return next The next token id for the collection
    /// @return max The maximum token id for the collection
    function getCollectionTokenIdRange(
        uint256 collectionId
    )
        public
        view
        collectionExists(collectionId)
        returns (uint256 min, uint256 next, uint256 max)
    {
        if (collections[collectionId].supply == 0) {
            return (
                _getGlobalTokenId(collectionId, 0),
                _getGlobalTokenId(
                    collectionId,
                    collectionStates[collectionId].nextId
                ),
                _getGlobalTokenId(collectionId, type(uint128).max)
            );
        }

        return (
            _getGlobalTokenId(collectionId, 0),
            _getGlobalTokenId(
                collectionId,
                collectionStates[collectionId].nextId
            ),
            _getGlobalTokenId(
                collectionId,
                collections[collectionId].supply - 1
            )
        );
    }

    // EVENTS ------------------------------------------------------------------

    /// @notice Emitted when when a collection is created or updated
    /// @param who The account that updated the collection
    /// @param collectionId The id of the collection that was updated
    event CollectionUpdate(address who, uint256 collectionId);

    /// MUTATIONS (INTERNAL) ---------------------------------------------------

    /// @notice "Use" up the next token ids for a given collection
    /// @dev Precondition: The collection for `collectionId` must exist
    /// @param collectionId The id of the collection
    /// @param count The number of token ids to consume
    function _consumeTokenIds(
        uint256 collectionId,
        uint256 count
    ) internal returns (uint256) {
        require(
            collectionStates[collectionId].nextId + count <=
                (
                    (collections[collectionId].supply > 0)
                        ? (collections[collectionId].supply)
                        : type(uint128).max
                ),
            "not enough supply"
        );
        uint256 firstTokenId = _getGlobalTokenId(
            collectionId,
            collectionStates[collectionId].nextId
        );
        collectionStates[collectionId].nextId += count;
        return firstTokenId;
    }

    /// MUTATIONS (PUBLIC) -----------------------------------------------------

    /// @notice Create a new collection
    /// @param collection The collection data
    function createCollection(
        Collection calldata collection
    ) public onlyCollectionAdmin(collections.length) {
        emit CollectionUpdate(msg.sender, collections.length);
        collections.push(collection);
        collectionStates.push(CollectionState(0));
    }

    /// @notice Update an existing collection
    /// @param collectionId The id of the collection to update
    /// @param collection The new collection data
    function updateCollection(
        uint256 collectionId,
        Collection calldata collection
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId] = collection;
    }

    /// @notice Update the paymentReceiver for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param paymentReceiver The new value for paymentReceiver
    function updateCollectionPaymentReceiver(
        uint256 collectionId,
        address paymentReceiver
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].paymentReceiver = paymentReceiver;
    }

    /// @notice Update the artist for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param artist The new value for artist
    function updateCollectionArtist(
        uint256 collectionId,
        address artist
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].artist = artist;
    }

    /// @notice Update the price for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param price The new value for price
    function updateCollectionPrice(
        uint256 collectionId,
        uint256 price
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].price = price;
    }

    /// @notice Update the friend price for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param price The new value for friend price
    function updateCollectionFriendPrice(
        uint256 collectionId,
        uint256 price
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].friendPrice = price;
    }

    /// @notice Update the supply for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param supply The new value for supply
    function updateCollectionSupply(
        uint256 collectionId,
        uint128 supply
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].supply = supply;
    }

    /// @notice Update the startTime for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param startTime The new value for startTime
    function updateCollectionStartTime(
        uint256 collectionId,
        uint256 startTime
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].startTime = startTime;
    }

    /// @notice Update the perMintQuantity for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param perMintQuantity The new value for perMintQuantity
    function updateCollectionPerMintQuantity(
        uint256 collectionId,
        uint256 perMintQuantity
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].perMintQuantity = perMintQuantity;
    }

    /// @notice Update the royaltyPercentage for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param royaltyPercentage The new value for royaltyPercentage
    function updateCollectionRoyaltyPercentage(
        uint256 collectionId,
        uint256 royaltyPercentage
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].royaltyPercentage = royaltyPercentage;
    }

    /// @notice Update the active value for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param active The new value for active
    function updateCollectionActive(
        uint256 collectionId,
        bool active
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].active = active;
    }

    /// @notice Update the directMint value for an existing collection
    /// @param collectionId The id of the collection to update
    /// @param directMint The new value for directMint
    function updateCollectionDirectMint(
        uint256 collectionId,
        bool directMint
    ) public collectionExists(collectionId) onlyCollectionAdmin(collectionId) {
        emit CollectionUpdate(msg.sender, collectionId);
        collections[collectionId].directMint = directMint;
    }
}

// File: contracts/interfaces/IERC2981.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.8.13;

import "@openzeppelin/contracts/utils/introspection/IERC165.sol";

///
/// @dev Interface for the NFT Royalty Standard
///
interface IERC2981 is IERC165 {
    /// ERC165 bytes to add to interface array - set in parent contract
    /// implementing this standard
    ///
    /// bytes4(keccak256("royaltyInfo(uint256,uint256)")) == 0x2a55205a
    /// bytes4 private constant _INTERFACE_ID_ERC2981 = 0x2a55205a;
    /// _registerInterface(_INTERFACE_ID_ERC2981);

    /// @notice Called with the sale price to determine how much royalty
    //          is owed and to whom.
    /// @param _tokenId - the NFT asset queried for royalty information
    /// @param _salePrice - the sale price of the NFT asset specified by _tokenId
    /// @return receiver - address of who should be sent the royalty payment
    /// @return royaltyAmount - the royalty payment amount for _salePrice
    function royaltyInfo(
        uint256 _tokenId,
        uint256 _salePrice
    ) external view returns (
        address receiver,
        uint256 royaltyAmount
    );
}

// File: contracts/interfaces/IOperatorFilterRegistry.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

interface IOperatorFilterRegistry {
    function isOperatorAllowed(address registrant, address operator) external view returns (bool);

    function register(address registrant) external;

    function registerAndSubscribe(address registrant, address subscription) external;

    function registerAndCopyEntries(address registrant, address registrantToCopy) external;

    function updateOperator(
        address registrant,
        address operator,
        bool filtered
    ) external;

    function updateOperators(
        address registrant,
        address[] calldata operators,
        bool filtered
    ) external;

    function updateCodeHash(
        address registrant,
        bytes32 codehash,
        bool filtered
    ) external;

    function updateCodeHashes(
        address registrant,
        bytes32[] calldata codeHashes,
        bool filtered
    ) external;

    function subscribe(address registrant, address registrantToSubscribe) external;

    function unsubscribe(address registrant, bool copyExistingEntries) external;

    function subscriptionOf(address addr) external returns (address registrant);

    function subscribers(address registrant) external returns (address[] memory);

    function subscriberAt(address registrant, uint256 index) external returns (address);

    function copyEntriesOf(address registrant, address registrantToCopy) external;

    function isOperatorFiltered(address registrant, address operator) external returns (bool);

    function isCodeHashOfFiltered(address registrant, address operatorWithCode)
        external
        returns (bool);

    function isCodeHashFiltered(address registrant, bytes32 codeHash) external returns (bool);

    function filteredOperators(address addr) external returns (address[] memory);

    function filteredCodeHashes(address addr) external returns (bytes32[] memory);

    function filteredOperatorAt(address registrant, uint256 index) external returns (address);

    function filteredCodeHashAt(address registrant, uint256 index) external returns (bytes32);

    function isRegistered(address addr) external returns (bool);

    function codeHashOf(address addr) external returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}